
##3 OTHER CONTROL STATEMENTS

The Indexed DO Statement

The following two segments of program code illustrate the indexed DO

statement.  The first one prints the integers 50,51,52,...,200, in

increasing order.  The second one prints multiples of 25 from 1000 down

to 100, inclusive, in decreasing order: 1000,975,950,...,125,100.

do K = 50, 200                                             (3)

     print*, K

end do

do count = 1000, 100, -25                                  (4)

    print*, count

end do

     The third value after the equals sign in an indexed DO statement

tells by how much the value of the control variable is to be changed on

each iteration of the loop; this quantity is called the STEP value.

If the third value is missing, 1 is added to the control variable on each

iteration (in other words, the STEP value defaults to 1).  The three

values can be any numeric expression, except that the STEP value cannot

be 0.  The control variable can be either of REAL or INTEGER type.

                                                                            17

EXERCISE 3.5:  How would you change code segment 3 to print all multiples

of 30 from 210 to 630, inclusive?

EXERCISE 3.6:  What statements do you write to print the squares of

all the even numbers from 10 to 80, inclusive?

     Example ##3A illustrates the basic logic for finding the smallest

of a sequence of numbers.  First you look at one number and note that as

the smallest you have seen so far.  Then you look at the other numbers in

the sequence and, at each one, replace your note of the smallest if the

one you are looking at is smaller than the one noted so far; if it is

not smaller than the one noted, do not make any change in it.  For

instance, if the input is 4, 7, 12, 5, 6, then the answer will be 5.

Analogous logic applies for finding the largest of a sequence of numbers.

EXAMPLE ##3A _________________________________________________

      program   read using the counter logic

C         Read a specified number of real values and tell what

C         the smallest of them is

      INTEGER number of values, count

      REAL  smallest so far, data

* BEGIN

      print*, 'Tell me how many numbers you want to enter'

      read*,  number of values

      print*, 'Now enter the first number'

      read*,  smallest so far

      do count = 2, number of values

          print*, 'Enter another number'

          read*, data

          if (data .lt. smallest so far) then

              smallest so far = data

          end if

      end do

      print*, 'The smallest number you entered was ',

     _        smallest so far

      end

____________________________________________________________

ANSWERS TO EXERCISES:====================================================

3.1:  Change the PROCESS statement to  \ sum = sum + data * data.

3.2:  Change the PROCESS statement to these three lines:

      \ if (mod (data,2) = 0)  \  sum = sum + data  \  end if

3.3:  The first value would not be added to the sum.  For instance, if

      the input were 12, 5, 8, 0, the sum would be reported as just 13.

      It would also add 0 to the sum, which does not hurt.

3.4:  The last value would not be added to the sum, and also an unknown

      value of data would be added on the first iteration.

=========================================================================

     The distinctive feature of logic for finding the smallest value is

that you must process the first value before you enter the loop.  If you

wanted to use Example ##3A to also find the sum of the data values being

read, you would (1) put the statement Sum=SmallestSoFar just before the

indexed DO, (2) insert Sum=Sum+Data after the READ statement inside the

loop, and (3) print Sum after the END DO.

     For an indexed DO loop, if the starting value is greater than the

ending value and the STEP value is positive, the indexed DO statement

does nothing but assign the starting value to the control variable.

Similarly, when the starting value is less than the ending value and the

STEP value is negative, the indexed DO statement does nothing but assign

the starting value to the control variable.

     In FORTRAN-77, you do not use the END DO statement.  Instead, you

put a CONTINUE statement in place of the END DO and put a statement label

(1 to 5 digits in columns 1 to 5) on that CONTINUE statement.  Also,

after the DO at the beginning of the loop, you put that same number.

For instance, using 22 as the label, the following two segments of code

are equivalent:

    do count = 1000, 100, -25                              (5)

        print*, count

    end do

    do 22 count = 1000, 100, -25                           (6)

        print*, count

22  continue

     VAX FORTRAN allows either form, but FORTRAN-77 just allows the

labeled form.  Actually, any executable statement is allowed in place of

CONTINUE; that statement is considered part of the body of the indexed DO

statement.  But it is considered good style to always use CONTINUE

statements to mark the point to which a jump is performed during

execution.  A CONTINUE statement can be used anywhere; it is classed as

executable even though it does not do anything.  Statement labels are

allowed on any statement in the program, as long as no two statements are

given the same label.

     You cannot change the value of the DO control variable inside the

loop.  You may change a variable appearing in the lower or upper limit

for the loop, but that will not change how many times the loop iterates.

EXERCISE 3.7:  What goes wrong with Example ##3A if the user at the

keyboard says he/she is entering only 1 number?

EXERCISE 3.8:  Explain why it is not a good idea to change .LT. to .LE.

in Example ##3A.

EXERCISE 3.9:  What changes would you make in Example ##3A so that it

computes and prints the LARGEST of all the numbers read?

EXERCISE 3.10: What difference would it make to change the first line in

Example ##3A containing DO to \ DO count = 1, number of values - 1

ANSWERS TO EXERCISES:====================================================

3.5:  replace the first line by  \ DO K = 210, 630, 30

3.6:  \ DO K = 10, 80, 2  \  PRINT*, K * K  \  END DO

=========================================================================

The GOTO Statement and ASSIGN Statements

The following are samples of the three kinds of GOTO statements available

in FORTRAN-77:

goto 30                                                    (7)

goto (30, 60, 99, 22, 15) N + 1

assign 99 to target

goto target

     The first statement has the effect of transferring control to the

statement labeled 30; thus statement 30 will be the next statement

executed.  The second statement transfers control to 30 if N+1 has the

value 1, to 60 if N+1 has the value 2, to 99 if N+1 has the value 3, and

so on (you always start counting from 1).  Any number of statement labels

can be used in this computed GOTO statement, and any numeric

expression can follow the parentheses.  The value of the expression is

converted using INT if it is REAL.  If the value is less than 1 or more

than the number of statement labels given, control is not transferred at

all (that is, the next statement executed is the one directly after that

computed GOTO statement).

     The GOTO Target statement transfers control to the statement whose

label has been assigned to Target by some previous statement.  This

allows you to change the target of the GOTO.  The variable (Target in

this case) must be declared as an INTEGER variable, and its value must

have been given to it using an ASSIGN statement, as in "ASSIGN 99 TO

Target".  However, giving it a value using ASSIGN makes it unusable for

storing INTEGERs, so be careful not to use a variable for INTEGERs if you

are using it for labels.  So the last two lines of code segment 7 will

next execute the line labeled 99 if N is less than 0 or more than 4.

The Special One-Line IF Statement

The LOGICAL IF statement is used in place of the BLOCK IF discussed

previously when the BLOCK IF has no ELSE or ELSE IF statements and has

only one statement after THEN.  The LOGICAL IF statement is always

written on just one line.  So the following statement is equivalent in

effect to the BLOCK IF in Example ##3A:

if (data .lt. smallest so far) smallest so far = data      (8)

The parentheses are required after the IF, and the single statement after

the parentheses cannot be another LOGICAL IF.

ANSWERS TO EXERCISES:====================================================

3.7:  Nothing, it works fine; the DO statement does nothing but Count=2.

3.8:  It would make no difference in the effect, but it may execute the

      assignment to Smallest So Far a few extra times, which needlessly

      wastes time.

3.9:  Replace Smallest by Largest throughout, then change .LT. to .GT.

3.10: No difference in effect (though Count ends up being 1 smaller).

=========================================================================

     FORTRAN does not have a CASE or REPEAT statement such as more modern

languages have.  An example of the use of LOGICAL IF and GOTO to simulate

the effect of the REPEAT statement is the following code segment, which

reads real numbers until a sentinel value of 0.0 is seen and then tells

what the product of those numbers was.  It is appropriate in situations

where you know there will always be at least one nonzero number to read.

    Answer = 1.0                                           (9)

    read*, Value

 5  continue

        Answer = Answer * Value

        read*, Value

    if (Value .ne. 0.0) goto 5

    print*, 'The product is ', Answer

     Example ##3B illustrates the use of the new kinds of statements

introduced in this section.  It reads 10 numbers from the keyboard,

using a standard method (called the stubborn method) to guarantee that

each number processed is at least 1 and at most 30.  For each number

read, the program tells the user something about it.

EXAMPLE ##3B _________________________________________________

      program   test numbers

C         Read 10 numbers in the range of 1 to 30 and tell

C         something interesting about each of them, when possible

      INTEGER  K, num, end of cases

      LOGICAL  boring number

* BEGIN

      boring number  = .false.

      do K  = 1, 10

  1       continue

              read*,  num

          if (num .lt. 1  .or.  num .gt. 30) goto 1

          assign 7 to end of cases

          goto (5, 4,6,4,6,6,3,4,6,4,5,

     _             6,6,4,6,3,6,6,6,4,5)  num - 9

              print*, 'It has 1 digit'

          goto end of cases

  3           print*, 'It is a square'

          goto end of cases

  4           print*, 'It is a 2-digit prime'

          goto end of cases

  5           print*, 'It is a multiple of 10'

          goto end of cases

  6           print*, 'It is not an interesting number'

              boring number  = .true.

  7       continue

      end do

      if (boring number) print*, 'Some of the numbers were boring.'

      end

____________________________________________________________

     Example ##3B illustrates a standard logic for finding out whether

none of the numbers in a sequence of numbers has a certain property.  To

find out whether none of the numbers is boring, we set a logical variable

to FALSE initially and then each time we see a number in the sequence, we

set the logical variable to TRUE if it has the property of being boring.

Additional Features of FORTRAN-77

Another kind of IF statement is the arithmetic IF\ROMAN' statement.

An example of it is IF(N-2)30,40,50.  It always has an expression within

parentheses followed by exactly three labels of statements.

     The effect of the statement IF(N-2)30,40,50 is that, if the

expression N-2 is negative, GOTO 30, else if N-2 is zero, GOTO 40, else

if N-2 is positive, GOTO 50.  This arithmetic IF statement is available

because the original 1957 form of FORTRAN was designed for a computer

that had a machine instruction that did such a three-way GOTO, and

FORTRAN-77 is designed to be upwardly compatible with the original

FORTRAN.  It should never be used in modern programs.

     The other control statements in FORTRAN include CALL and RETURN,

which are discussed in the next section of this book.  The syntax of some

control statements is given by the following EBNF definitions.  One more

kind of notation is used:  material enclosed in the brackets [ ] is

optional (you can put it in or leave it out).

$  statement  =  DO  identifier  =  expression  ,  expression

                            [  ,  expression  ] CR

                     { statement }  END DO CR

              |  DO label identifier = expression , expression

                            [  ,  expression  ] CR

                     { statement }  label CONTINUE  CR

              |  IF   (  expression  )  THEN  CR

                            { statement }

                     {  ELSE IF   (  expression  )  THEN CR

                            { statement }  }

                     [  ELSE  CR   { statement } ]

                     END IF  CR

              |  IF   (  expression  )  statement   CR

              |  IF   (  expression  )  label, label, label  CR

              |  GOTO label  CR

              |  GOTO  (  label {  , label }  )  expression  CR

              |  GOTO identifier  CR

              |  ASSIGN label TO identifier  CR

              |  CONTINUE  CR

              |  END  CR

     For the indexed DO with label, the label on the corresponding

CONTINUE statement must be the same label.  The expression after IF must be a LOGICAL expression, but the expressions in the indexed

DO and the computed GOTO must be NUMERIC expressions.

